
Abstract. Mobile ad hoc networks (MANETs) are increasingly popular
and deployed in a wide range of environments. However, it is challenging
to formally analyze a MANET, both because there are few reasonably
accurate formal models of mobility, and because the large state space
caused by the movements of the nodes renders straightforward model
checking hard. In particular, the combination of wireless communication
and node movement is subtle and does not seem to have been adequately
addressed in previous formal methods work. This paper presents a for-
mal executable and parameterized modeling framework for MANETs in
Real-Time Maude that integrates several mobility models and wireless
communication. We illustrate the use of our modeling framework with the
Ad hoc On-Demand Distance Vector (AODV) routing protocol, which
allows us to analyze this protocol under different mobility models.

1 Introduction

A mobile ad hoc network (MANET) is a self-configuring network of mobile
devices (laptops, smart phones, sensors, etc.) that communicate wirelessly and
cooperate to provide the necessary network functionality. Since MANETs can
form ad hoc networks without fixed infrastructure, they are supposed to have
a wide applicability, for example for providing ad hoc networks for cooperating
“smart” cars, for emergency responders during accidents, during natural disas-
ters which may disable fixed infrastructure, in battlefield areas, and so on.

Although many such applications are safety-critical and need formal analysis
to ensure their correctness, the formal modeling and analysis of MANETs present
a number of challenges that include:

1. The need to model node movement realistically.
2. Modeling communication. There is a subtle interaction between wireless com-

munication, which typically is restricted to distances of between 10 and
100 m, and node mobility. For example, nodes may move into, or out of, the
sender’s transmission range during the communication delay; furthermore,
the sender may itself move during the communication. Modeling communi-
cation in MANETs is therefore challenging for formal languages, which are
usually based on fixed communication primitives.

c© Springer International Publishing Switzerland 2014
S. Escobar (Ed.): WRLA 2014, LNCS 8663, pp. 162–177, 2014.
DOI: 10.1007/978-3-319-12904-4 9



A Framework for Mobile Ad hoc Networks in Real-Time Maude 163

3. Since the communication topology of the network depends on the locations
of the nodes, such locations must be taken into account in the model. How-
ever, this leads to very large state spaces, which makes direct model checking
analysis unfeasible: if there are m nodes and n locations, there are nm differ-
ent node/location-states. A 10×10 grid with four nodes would therefore lead
to 100 million states just to capture all nodes and their locations.

As explained in Sect. 7, we are not aware of any formal model that provides a
reasonably detailed model of both mobility and communication in MANETs.
Because of its expressiveness and flexibility to define models of communication,
Real-Time Maude [20] is a promising language for formally modeling MANETs.
In this paper we provide, to the best of our knowledge, the first reasonably
detailed formal modeling framework for MANETs. In particular, we formalize

– the most popular models for node mobility, and
– geographically bounded wireless communication, which takes into account the

interplay between communication delay and mobility,

in Real-Time Maude. Furthermore, we use object-oriented techniques to make
it easy to compose our framework with a model of a MANET protocol.

Concerning Challenge 3 above, in this paper we do not develop abstrac-
tion techniques for node mobility. Instead, to be able to perform model check-
ing analysis, our model is parametric in aspects such as the possible velocities
and directions a node can choose. However, even if a node moves slowly, it
may still cover the entire area (and hence contribute to an unmanageable state
space) given enough time. Another key feature of Real-Time Maude that makes
some meaningful model checking analysis of MANETs possible is therefore time-
bounded model checking, which allows us to analyze scenarios only up to a certain
duration (during which the nodes may not reach most locations). Abstracting the
state space caused by node mobility and the need to keep track of node locations
is the sine qua non for serious model checking of MANETs. The point is that
this paper lays the foundations for developing such abstractions by providing a
first reasonably detailed formal model of location-aware MANETs.

One of the main tasks of a MANET is to maintain an (ad hoc) network, which
means that the network must figure out how to route messages between nodes.
In this paper we illustrate the use of our MANETs framework by modeling and
analyzing the widely used Ad hoc On-Demand Distance Vector [22] (AODV)
routing protocol for MANETs developed by the IETF MANET working group.

The rest of this paper is organized as follows. Section 2 gives a background to
Real-Time Maude. Section 3 briefly introduces MANETs. Section 4 presents our
Real-Time Maude modeling framework for MANETs. Section 5 shows how our
framework can be used to model the AODV protocol, and Sect. 6 explains how
that model of AODV can be model checked using Real-Time Maude. Finally,
Sect. 7 discusses related work and Sect. 8 gives some concluding remarks.

Due to space limitations, we have to omit many details; they are all given in
our accompanying longer report [14].

2 Real-Time Maude

Real-Time Maude [20] is a language and tool that extends Maude [5] to support
the formal specification and analysis of real-time systems.

Specification. A Real-Time Maude module specifies a real-time rewrite theory
(Σ,E ∪ A, IR,TR), where:

– Σ is an algebraic signature; that is, a set of declarations of sorts, subsorts,
and function symbols.

– (Σ,E∪A) is a membership equational logic theory [5], with E a set of possibly
conditional equations, and A a set of equational axioms such as associativity,
commutativity, and identity. (Σ,E ∪ A) specifies the system’s state space as
an algebraic data type, and includes a specification of a sort Time.

– IR is a set of labeled conditional rewrite rules specifying the system’s local
transitions, each of which has the form1 [l] : t −→ t′ if

∧m
j=1 cond j , where

each cond j is either an equality uj = vj or a rewrite tj −→ t′j , and l is a label.
Such a rule specifies an instantaneous transition from an instance of t to the
corresponding instance of t′, provided the condition holds.

– TR is a set of tick rules l : {t} −→ {t′} in time τ if cond that advance time
in the entire state t by τ time units.

A class declaration class C | att1 : s1, . . . , attn : sn declares a class
C with attributes att1 to attn of sorts s1 to sn. An object of class C in a given
state is represented as a term <O : C | att1 : val1, ..., attn : valn > of sort Object,
where O, of sort Oid, is the object’s identifier, and where val1 to valn are the
current values of the attributes att1 to attn. A message is a term of sort Msg.

The state of an object-oriented specification is a term of sort Configuration,
and is a multiset of objects and messages. Multiset union is denoted by an
associative and commutative juxtaposition operator, so that rewriting is multiset
rewriting. For example, the rewrite rule

rl [l] : m(O,w) < O : C | a1 : x, a2 : O’, a3 : z >

=>

< O : C | a1 : x + w, a2 : O’, a3 : z > dly(m’(O’,x), z) .

defines a family of transitions in which a message m, with parameters O and w,
is read and consumed by an object O of class C, the attribute a1 of object O
is changed to x + w, and a new message dly(m’(O’,x),z) is generated; this
message will become the “ripe” message m’(O’,x) after z time units. Attributes
whose values do not change and do not affect the next state of other attributes
or messages, such as a3, need not be mentioned in a rule. Attributes that are
unchanged, such as a2, can be omitted from right-hand sides of rules.

A subclass inherits all the attributes and rules of its superclasses.
1 An equational condition ui = vi can also be a matching equation, written ui := vi,

which instantiates the variables in ui to the values that make ui = vi hold, if any.

Formal Analysis. In this paper, we only consider Real-Time Maude’s linear
temporal logic model checker, which analyzes whether each behavior satisfies
a temporal logic formula. State propositions are terms of sort Prop, and their
semantics is defined by equations ceq statePattern |= prop = b if cond, for
b a term of sort Bool, stating that prop evaluates to b in states that are instances
of statePattern when the condition cond holds. These equations together define
prop to hold in all states t where t |= prop evaluates to true. A temporal
logic formula is constructed by state propositions and temporal logic operators
such as True, False, ~ (negation), /\, \/, -> (implication), [] (“always”), <>
(“eventually”), and U (“until”). Real-Time Maude provides both unbounded and
time-bounded LTL model checking. The time-bounded model checking command

(mc t |=t formula in time <= timeLimit .)

checks whether the temporal logic formula formula holds in all behaviors up to
duration timeLimit starting from the initial state t.

3 Mobility and Communication Delay in MANETs

This section gives an overview of the main mobility models used by researchers
on protocol evaluations, and of the per-hop delay in wireless communication.

Fig. 1. Motion paths of a mobile node in three mobility models, where a bullet • depicts
a pause in the movement.

Mobility Models. Different mobility patterns have been proposed to model node
mobility in realistic scenarios. In this paper we focus on the following main
entity mobility models [2], also illustrated in Fig. 1, in which a node’s movement
is independent of the movements of the other nodes:

– Random Walk: Each node moves in “rounds” of fixed durations. A node moves
in the same direction and with the same speed throughout one round. At the
end of each round, the new speed and the new direction of a node are randomly
chosen, and a new moving round starts.

– Random Waypoint: Each node initially pauses for a fixed duration. When a
pause ends, a node randomly chooses a new destination and a new speed, and
then travels to that destination at the chosen speed. After arriving, the node
again pauses before a new moving round starts.

– Random Direction: Each mobile node chooses a random direction, along which
it travels until reaching the border of the sensing area. When a node arrives
at the border, the node pauses for a given time, and then randomly selects a
new direction and starts to move in that direction.

Communication Delay. To understand how node movement affects wireless com-
munication, we must understand the messaging delays in wireless communica-
tion. In a typical wireless transmit/receive process, the per-hop communication
delay from a transmitter to a receiver consists of the following five phases [24]:

Delay factor Description

Sender processing delay The time elapsed on the sender side from the moment a
message timestamp is taken to the point the message is
buffered in the device

Media access delay The time for a message to stay in the radio device buffer;
e.g., in a CSMA system, this is the delay waiting for a clear
channel to transmit

Transmit delay The time for a radio device to transmit a message over a
radio link

Radio propagation delay The time for a message to propagate through the air to a
receiver

Receiver processing delay The time spent on the receiver side to pass the received
message from the device buffer to the application module

We can abstract from the radio propagation delay, since the transmission range
in MANETs typically ranges from 10 to 100 m, while the radio propagation
speed is approximately 3 × 108 m/s. The media access delay depends on the
MAC overhead, such as collisions and waiting time.

4 Formalizing MANET Mobility and Communication

This section presents a modeling framework for MANETs with nodes that com-
municate wirelessly. Section 4.2 shows how mobile nodes can be specified in
Real-Time Maude, Sect. 4.3 explains how the timed behavior of MANETs can
be defined in a way that allows us to easily compose our model with MANET
protocols, and Sect. 4.4 formalizes wireless communication for MANETs.

4.1 Some Basic Data Types

We assume a sort Location for the set of locations, a sort Speed for the different
velocities with which a node can move, a set Direction for the different direc-
tions that a node can choose, and sorts SpeedRange, DirRange, and DestRange
denoting sets of, respectively, Speed, Direction, and Location elements.

We also assume that nodes move in a two-dimensional square with length
areaSize. A location is therefore represented as a pair x .y of rational numbers:2

op _._ : Rat Rat ~> Location [ctor] .

cmb X . Y : Location if 0 <= X and X <= areaSize /\ 0 <= Y and Y <= areaSize .

We do not further specify the different powersets, whose elements could be
unions of dense intervals or of single points, or both. Since the nodes need to
nondeterministically select a new speed, a new next destination, and/or a new
next direction, we assume for generality’s sake that there is an operator choose
that can select any value in the respective set nondeterministically, and an oper-
ator [ ], so that an element e can be chosen from a set S if and only if there is
a rewrite (in zero or more steps) choose(S ) => [e ]. For example, if we have a
discrete set of possible next directions d1 ; d2 ; ...; dn, where the set union
operator ; is declared to be associative and commutative, we can specify that
any value from the set can be selected, by giving the following rewrite rule:

var D : Direction . var DR : DirRange .

rl [chooseDir] : choose(D ; DR) => [D] .

4.2 Modeling Mobile Nodes

We model a MANET node in an object-oriented style, where a mobile node is
modeled as an object instance of some subclass of the following base class Node:

class Node | currentLocation : Location .

The attribute currentLocation denotes the node’s current location. A station-
ary node is an object instance of the subclass StationaryNode that does not
add any attribute to Node:

class StationaryNode . subclass StationaryNode < Node .

A mobile node is modeled as an object of a subclass of the class MobileNode:

class MobileNode | speed : Speed, direction : Direction, timer : TimeInf .

subclass MobileNode < Node .

where speed and direction denote, respectively, the node’s current speed and
its current movement direction. The timer attribute is used to ensure that a
node changes its movement (or lack thereof) in a timely manner; that is, timer
denotes the time remaining until some discrete event must take place.
2 We do not show most variable declarations, but follow the Maude convention that

variables are written in capital letters.

Random Walk. A node moving according to the random walk model is continu-
ously moving, in time intervals of length movingTime. At the end of an interval,
the node nondeterministically chooses a new speed and a new direction for its
next interval. Such a node is modeled by an object of the subclass RWNode:

class RWNode | speedRange : SpeedRange, dirRange : DirRange .

subclass RWNode < MobileNode .

where speedRange and dirRange denote the set of possible next speeds and
directions, respectively. The timer attribute inherited from its superclass denotes
the remaining time of its current move interval. The instantaneous behavior of
the mobility part of such a node can be modeled by the following rule. In this
rule, the node is finishing one interval (the timer attribute is 0), and must select
new speed and direction for its next round, and reset the timer:

crl [startNewMove] :

< O : RWNode | timer : 0, speedRange : SR, dirRange : DR >

=>

< O : RWNode | timer : movingTime, speed : S, direction : D >

if choose(SR) => [S] /\ choose(DR) => [D] .

The actual movement of such a node is modeled in Sect. 4.3.

Random Waypoint. In the random waypoint mobility model, a node alternates
between pausing and moving. When it starts moving, it selects a new speed and
a new destination and starts moving towards the destination. Such a node should
be modeled by an object instance of the RWPNode subclass:

class RWPNode | speedRange : SpeedRange, destRange : DestRange,

status : Status .

subclass RWPNode < MobileNode .

The status attribute is either pausing or moving, and destRange denotes the
possible next destinations.

The instantaneous behavior of this mobility model is given by the following
rewrite rules. First, if the node is pausing and the timer expires, the node must
get moving by selecting a new speed and desired next location, and resetting the
timer so that it expires when the goal location is reached:

var MOVE-TIME : Time .

crl [startMoving] :

< O : RWPNode | currentLocation : CURR-LOC, status : pausing,

timer : 0, speedRange : SR, destRange : DER >

=>

< O : RWPNode | status : moving, speed : S,

direction : D, timer : MOVE-TIME >

if choose(SR) => [S] /\ choose(DER) => [NEXT-LOC]

/\ D := direction(L, NEXT-LOC)

/\ MOVE-TIME := timeBetweenLocations(CURR-LOC, NEXT-LOC, S) .

where direction gives the direction from one location to another, and time-
BetweenLocations denotes the time it takes to travel between two locations at a
given speed. The selected speed cannot be zero, unless the selected next location
is also the current location, because then the last matching equation would not
hold, since the traveling time between the two locations would be the infinity
value INF, which is not a Time value.

The following rule applies when the timer of a moving node expires; then it
is time to take a rest for pauseTime time units:

rl [startPausing] :

< O : RWPNode | status : moving, timer : 0 >

=>

< O : RWPNode | status : pausing, timer : pauseTime, speed : 0 > .

Random direction nodes can be defined in the same way; see [14] for details.

4.3 Timed Behavior and Compositionality

Our model of mobile nodes must be easily composable with “application” pro-
tocols such as AODV to define a particular MANET system. The straightfor-
ward way of composing our model of mobility with a MANET protocol is to
let the nodes in the application protocol be modeled as objects of subclasses of
the classes introduced above, since a subclass “inherits” all the attributes and
rewrite rules of its superclasses; in particular, such application-specific subclasses
would inherit the rewrite rules modeling the movements of their nodes.

However, we must allow the user to define the timed behavior of her system,
and compose it with the timed behavior of mobile nodes. We therefore use the
following extension of the “standard” tick rule for object-oriented specifications:

var T : Time . var C : Configuration .

crl [tick] : {C} => {timeEffect(timeEffectMob(C, T), T)} in time T

if T <= min(mte(C), mteMob(C)) .

where timeEffectMob defines the effect of time elapse on the mobility-specific
parts of the system, and timeEffect defines how the passage of time changes the
state in the other parts of the composed system. Likewise, mteMob denotes the
maximum amount of time that may elapse from a given state until some mobility
action must be taken, and mte defines the amount of time until the application
protocol must perform a discrete action. These functions distribute over the
objects and messages in the configuration as explained in [14].

Since the speed is 0 when a node is pausing, we can easily define the timed
behavior of both stationary and mobile nodes. First of all, time does not affect
(the mobility-specific parts of) a stationary node:

eq timeEffectMob(< O : StationaryNode | >, T) = < O : StationaryNode | >.

Time affects a mobile node by moving the node and decreasing its timer value:

eq timeEffectMob(< O : MobileNode | currentLocation : L, speed : S,

direction : D, timer : T1 >, T)

= < O : MobileNode | currentLocation : move(L,S,D,T),

timer : T1 monus T > .

where move(l,s,d,t) denotes the location resulting from moving a node in loca-
tion l for t time units in direction d and with speed s. This function also makes
sure that a node does not move beyond the area under consideration.

The mobility model does not restrict the time advance for stationary nodes,
whereas for mobile nodes, time can advance until the timer becomes 0:

eq mteMob(< O : StationaryNode | >) = INF .

eq mteMob(< O : MobileNode | timer : T >) = T .

4.4 Modeling Wireless Communication in Mobile Systems

Only nodes that are sufficiently close to the sender, i.e., within the sender’s
transmission range, receive a message with sufficient signal strength. However,
both the sender and the potential receivers might move (possibly out of, or into,
the sender’s transmission range) during the entire communication delay.

As mentioned in Sect. 3, the total communication “delay” can be decomposed
into five parts. If we abstract from the radio propagation delay, the per-hop delay
can be seen to consist of two parts: the delay at the sender side (including sender
processing delay, media access delay, and transmit delay) and the delay at the
receiver side (including receiver processing delay). The point is that exactly those
nodes that are within the transmission range of the sender when the sending delay
ends should receive a message.

It is also worth mentioning that our model is still somewhat abstract and
does not capture all network factors, most notably collisions.

In MANETs communication can be by broadcast, unicast, or groupcast,
depending on which kind of message a transmitter intends to send, and who
are the recipients. In our model we have three corresponding message construc-
tors for broadcast, unicast, and groupcast, respectively:

msg broadcast_from_ : MsgContent Oid -> Msg .

msg unicast_from_to_ : MsgContent Oid Oid -> Msg .

msg gpcast_from_to_ : MsgContent Oid NeighborSet -> Msg .

When a node sender wants to broadcast some message content mc, it gener-
ates a “message” broadcast mc from sender. The following equation adds the
delay on the sending side, sendDelay, to this “broadcast message:”

eq broadcast MC from O = dly(transmit MC from O, sendDelay) .

The crucial moment is when the sending delay expires and the transmit
message becomes “ripe.” All the nodes that are within the transmission range
of the sender at that moment should receive the message. This distribution is

performed by the function distrMsg, where distrMsg(snd, loc, mc, conf ) gen-
erates a single message, with content mc, to each node in conf that is currently
within the transmission range of location loc; furthermore, this single message
has delay recDelay, modeling the delay at the receiving site:

eq {< O : Node | currentLocation : L > (transmit MC from O) C}

= {< O : Node | > distrMsg(O, L, MC, C)} .

eq distrMsg(O, L, MC, < O’ : Node | currentLocation : L’ > C)

= < O’ : Node | currentLocation : L’ > distrMsg(O, L, MC, C)

(if L withinTransRangeOf L’ then dly((MC from O to O’), recDelay)

else none fi) .

Unicast and groupcast are modeled similarly.

5 Case Study: Route Discovery in AODV

This section first gives an overview of the AODV routing protocol, and then
presents our Real-Time Maude model of AODV, focusing on the route discovery
process. The entire executable Real-Time Maude specification is available at
http://www.ifi.uio.no/RealTimeMaude/MANET/wrla2014-manets.rtmaude.

5.1 Route Discovery in AODV

AODV [22] is a widely used algorithm for routing messages between mobile nodes
which dynamically form an ad hoc network. AODV allows a source node to
initiate a route discovery process on an on-demand basis to establish a route to
a destination node.

Fig. 2. Route discovery process.

A source node S initiates a route dis-
covery process by broadcasting a route
request (RREQ) message to its neigh-
bors. An intermediate node can either
unicast a route reply (RREP) message
back to the source if a valid route to
the destination D can be found in its
local routing table, or re-broadcast the
received RREQ to its own neighbors. As
the RREQ travels from S to D, reverse
paths from all nodes back to S are auto-
matically set up. Eventually, when the
RREQ reaches D, it sends a RREP back
along the previously established reverse

path. After this process, a route between S and D is set up.

http://www.ifi.uio.no/RealTimeMaude/MANET/wrla2014-manets.rtmaude

5.2 Modeling Route Discovery in Real-Time Maude

Modeling Nodes and Messages. We model an AODV node as an object of a
subclass AODVNode of class Node. The new attributes show the identification of
a node’s routing request, the sequence number of a node itself, the local routing
table, and the buffered routing requests sent since the beginning of the current
round, respectively.

class AODVNode | rreqID : Nat, sequenceNumber : Sqn,

routingTable : RouteTable, requestBuffer : RreqBuffer .

subclass AODVNode < Node .

A routing table of sort RouteTable is modeled using the predefined data type
MAP. It consists of routing table entries of the form Oid |-> Tuple3, mapping
a destination node Oid to a 3-element tuple: the next hop towards the destina-
tion, the distance to the destination, and the local destination sequence number.
A route request buffer of the sort RreqBuffer is specified as a set of requests,
each of which is of the form Oid ∼ Sqn, and uniquely identifies a route request
by the identifier of a node and its sequence number.

In the AODV route discovery process there are mainly two kinds of messages:
RREQ and RREP. They are specified in our model as rreq(...) and rrep(...)
respectively. The message content will be illustrated below.

Modeling Route Discovery. A route discovery process in AODV consists of three
parts: initiating route discovery, route request handling, and route reply han-
dling. We only illustrate part of the route request handling, and refer the reader
to our longer report [14] for more details.

The RREQ-handling rules specify all events that may happen when a route
request is received. The receiving node first checks whether a received (OIP ∼
RREQID) has already been stored locally in the request buffer. If so, the route
request is ignored and the local routing table is updated by adding a routing
table entry towards the sender; otherwise, the receiving node adds the new route
request identifier to the request buffer, and takes further actions according to
the roles played by the receiving node. In the following case, the receiving node
is an intermediate node.

When receiving the RREQ message, an intermediate node either: (a) gen-
erates a route reply to the sender, or (b) re-broadcasts the received RREQ to
its neighbors. For example, action (a), as the following rewrite rule shows, hap-
pens only when O’s local information is fresher than that in the RREQ message
(DSN <= localdsn(RT[DIP])). Then O unicasts the route reply with the fresher
destination sequence number and its distance in hops from the destination along
the route back to the source node.

crl [on-receiving-rreq-3] :

(rreq(OIP,OSN,RREQID,DIP,DSN,HOPS,SIP) from SIP to O)

< O : AODVNode | routingTable : RT, requestBuffer : RB >

=>

< O : AODVNode | routingTable : RT’’,

requestBuffer : (OIP ∼ RREQID, RB) >

(msg rrep(OIP,DIP,localdsn(RT’’[DIP]),hops(RT’’[DIP]),O)

from O to nexthop(RT’’[OIP]))

if RT’ := update(SIP,SIP,1,0,RT) /\

RT’’ := update(OIP,SIP,HOPS + 1,OSN,RT’) /\

not (OIP ∼ RREQID) in RB /\ inRT(RT,DIP) /\

DIP =/= O /\ DSN <= localdsn(RT[DIP]) .

6 Formal Analysis of AODV

In this section we analyze the AODV route discovery process under different
mobility models. We therefore define node objects that belong to a subclass of
both AODVNode and a class defining the desired mobility pattern. For example,
a node moving according to the random waypoint model is an object instance
of the class RWPANode, and a stationary node is an instance of the class SANode:

class RWPANode . subclass RWPANode < RWPNode AODVNode .

class SANode . subclass SANode < StationaryNode AODVNode .

The main objective of a routing protocol such as AODV is that a route
between the desired source and the desired destination is eventually estab-
lished. To analyze this property, we define a parameterized atomic proposition
route-found(SRC,DEST) to hold if we can find, in the routing table of the source
node SRC, a routing table entry towards the destination node DEST:

op route-found : Oid Oid -> Prop [ctor] .

eq {< SRC : AODVNode | routingTable : RT , (DEST |-> TP) > REST}

|= route-found(SRC, DEST) = true .

The desired property of AODV can then be formalized as the temporal logic
formula <> route-found(...). Given an initial state initConfig, the follow-
ing command returns true if our desired property holds in the first test round
(roundTime); otherwise, a trace showing a counterexample is provided.

(mc {initConfig} |=t <> route-found(src,dest) in time <= roundTime .)

Experiment Scenarios. We define the following setting for our experiments:

– The transmission range is 10m, and the test area is 100m × 100m.
– The test round is 100s. The delay at the sender and at the receiver is set to

10s and 5s, respectively.
– The range of possible velocities is the singleton set (1).
– Nodes can move right, up, left or down: the direction range is a subset of
(0,90,180,270), and the destination range is a subset of four locations in
the corresponding four directions based on a node’s current location.

We have analyzed AODV in seven different scenarios; five of them are described
below and the other cases are described in our longer report [14]:

– Scenario (i), shown in Fig. 2, has five stationary nodes, where node 1, located
at (45 . 45), wants to build a route to node 5, located at (60 . 50), and nodes
2, 3 and 4 are at (50 . 50), (50 . 40), and (60 . 40), respectively.

– Scenario (i’), shown in Fig. 3 (a solid circle refers to the initial location of a
node, while a dashed circle refers to some point along the motion path of a
node), has the same topology as Scenario (i), but now node 2 is a random
waypoint node that can move up. We set its pause time to: (a) 10s, (b) 30s,
or (c) 60s. The initial state of this scenario is specified as:

eq src = 1 .

eq initConfig =

(bootstrap src)

< 1 : SANode | currentLocation : 45 . 45 , rreqID : 10, sequenceNumber : 1,

routingTable : empty, requestBuffer : empty >

< 2 : RWPANode | currentLocation : 50 . 50, speed : 0, direction : 0, timer : pauseTime,

speedRange : (1), destRange : (50 . 60), status : pausing, rreqID : 20,

sequenceNumber : 1, routingTable : empty, requestBuffer : empty >

< 3 : SANode | currentLocation : 50 . 40, ... >

< 4 : SANode | currentLocation : 60 . 40, ... >

< 5 : SANode | currentLocation : 60 . 50, ... > .

– Scenario (ii), also shown in Fig. 3, has three nodes with both nodes 2, located
at (40 . 50), and 3 (a random waypoint node located at (50 . 40)) intending
to build a route to the destination node 1 located at (50 . 50).

Fig. 3. Scenarios (i’) and (ii)

Analysis Results. The results of the model checking show that the desired prop-
erty holds in Scenarios (i), (i’)-(a), and (i’)-(c), but not in Scenarios (i’)-(b)
and (ii).

In Scenario (i’)-(b), the pause time (30s) allows node 2 to forward the RREQ
message from node 1 to node 5. However, node 2 cannot receive the RREP

message from node 5 due to its movement (the dash circle in this case is at
(50 . 60)). Meanwhile, since node 5 has already recorded node 1’s RREQ from
node 2, it ignores the one from node 4.

In Scenario (ii), before sending out the RREQ message, node 3 moves left
to a new location (40 . 40) within the transmission range of node 2. Thus, to
establish the route to node 1, node 3’s RREQ message needs to be forwarded
by node 2. However, the model checking counterexample shows that route dis-
covery for node 3 fails: no route can be found between nodes 3 and 1, though
obviously node 2 succeeds in building a route to node 1. This problem arises
due to the discarding of the RREP message. As stated in [22], an intermediate
node forwards a RREP message only if the RREP message serves to update its
routing table entry towards the destination. However, in this case, node 2 has
already secured an optimal route to node 1 before receiving the RREQ message
from node 3. Fehnker et al. [7] also pointed out this problem, but in a static
linear topology with three nodes.

7 Related Work

There are a number of formal specification and analysis efforts of MANETs in
general, and AODV in particular.

Bhargavan et al. [1] use the SPIN model checker to analyze AODV. They
only consider a 3-node topology with one link break, but without node move-
ment, and communication delay is not considered. Chiyangwa et al. [4] apply
the real-time model checker Uppaal to analyze AODV. They only consider a
static linear network topology. Although they take communication delay into
account, the effect of mobility on communication delay is not considered, since
the topology is fixed. Fehnker et al. [7] also use Uppaal to analyze AODV. They
also only considered static topologies, or simple dynamic topologies by adding or
removing a link, and those topologies are based on the connectivity graph with-
out concrete locations for nodes. Furthermore, no timing issues are considered.
Höfner et al. [12] apply statistical model checking to AODV. However, mobility is
simply considered by arbitrary instantaneous node jumping between zones that
split the whole test grid. Although they take into account the communication
delay, the combination of mobility and communication delay is not considered.
None of these studies has built a generic framework for MANETs. Our modeling
framework aims at the combination of wireless communication and mobility, and
allows formal modeling and analysis of protocols under realistic mobility models.

On the process algebra side, CWS [17], CBS# [19], CMAN [10], CMN [15],
the ω-calculus [23], RBPT [9,11], TCWS [16] and AWN [6], have been proposed
as process algebraic modeling languages for MANETs. These languages feature
a form of local broadcast, in which a message sent by a node could be received
by other nodes “within transmission range.” However, the connectivity is only
considered abstractly and logically, without taking into account concrete loca-
tions and transmission range for nodes. Furthermore, [17] only considers fixed
network topologies, whereas the others (except [11]) deal with arbitrary changes

in topology. Godskesen et al. [11] consider realistic mobility, and propose con-
crete mobility models. However, no protocol application or automated analysis
is given, and communication delay is not taken into account. Merro et al. [16]
propose a timed calculus with time-consuming communications, and equip it
with a formal semantics to analyze communication collisions.

Generally, these studies have proposed a framework for MANETs, but they
lack of either mobility modeling or timing issues handling.

There are also a number of well known “ambient” calculi for mobility, such as
the ambient calculus [3], the π-calculus [18], and the join-calculus [8]. However,
these are very abstract models that do not take locations and geographically
bounded communication into account, and are therefore not suitable to model
MANETs at the level of abstraction considered in this paper.

Finally, Maude and Real-Time Maude have been applied to analyze wire-
less sensor networks, but the work in [13,21] does not consider node mobility
(even though [13] mentions that mobility is addressed in a technical report in
preparation; however, we cannot find that technical report).

8 Concluding Remarks

We have defined in Real-Time Maude what we believe is the first formal model of
MANETs that provides a reasonably faithful model of popular node movement
patterns and wireless communication. We have used our compositional model to
specify and formally analyze the AODV routing protocol, and have shown that
such Real-Time Maude analysis could easily find the known flaw in AODV.

We have abstracted from message collision, which should also be considered in
our model. The price to pay for having a much more realistic model of MANETs
than other formal approaches is that the state space quickly becomes too large
for model checking. We should therefore develop statistical model checking tech-
niques for MANETs. Most importantly, we should develop abstraction tech-
niques for MANETs. The formalization presented in this paper has provided the
necessary foundation for such efforts.

Acknowledgments. We thank the anonymous reviewers for helpful comments on a
previous version of this paper. This work has been partially supported by AFOSR
Contract FA8750-11-2-0084 and NSF Grant CNS 13-19109.

References

1. Bhargavan, K., Obradovic, D., Gunter, C.: Formal verification of standards for
distance vector routing protocols. J. ACM 49(4), 538–576 (2002)

2. Camp, T., Boleng, J., Davies, V.: A survey of mobility models for ad hoc network
research. Wirel. Commun. Mob. Comput. 2(5), 483–502 (2002)

3. Cardelli, L., Gordon, A.D.: Mobile ambients. In: Proceedings of POPL’98. ACM
(1998)

4. Chiyangwa, S., Kwiatkowska, M.: A timing analysis of AODV. In: Steffen, M.,
Zavattaro, G. (eds.) FMOODS 2005. LNCS, vol. 3535, pp. 306–321. Springer,
Heidelberg (2005)



A Framework for Mobile Ad hoc Networks in Real-Time Maude 177

5. Clavel, M., Durán, F., Eker, S., Lincoln, P., Mart́ı-Oliet, N., Meseguer, J., Talcott,
C. (eds.): All About Maude. LNCS, vol. 4350. Springer, Heidelberg (2007)

